This notebook has been saved with all outputs deleted. To recreate the output keyboard Alt+v o; and wait until the notebook has been evaluated, 20 seconds on my office PC.

Example 1

Solve the following PDE

$$t\,\frac{\partial u}{\partial x} + 3\,\frac{\partial u}{\partial t} = -u, \quad u(x, 0) = f(x).$$

The vector field defined by this equation is $< t, 3, -z >$. Let us picture this field

In[318]:=
```
vecs1 = VectorPlot3D[{t, 3, -z}, {x, -2 Pi, 2 Pi}, {t, 0, 10}, {z, -1.1, 1.1},
  BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}]
```

Out[318]=



In[319]:=
```
Clear[sol1];

sol1[s_, r_] =
 FullSimplify[{x[s], t[s], z[s]} /. DSolve[{x'[s] == t[s], t'[s] == 3, z'[s] == -z[s],
      x[0] == r, t[0] == 0, z[0] == Cos[r]}, {x[s], t[s], z[s]}, s][[1]]]
```

Out[320]= $\left\{r + \frac{3\,s^2}{2},\ 3\,s,\ e^{-s}\,\mathrm{Cos}[r]\right\}$

In[321]:=
```
ParametricPlot[Evaluate[Table[sol1[s, r][[{1, 2}]], {r, -10, 10, .2}]],
 {s, 0, 10}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}}]
```

Out[321]=



In[322]:=
```
Simplify[Solve[{x == sol1[s, r][[1]], t == sol1[s, r][[2]]}, {s, r}][[1]]]
```

Out[322]= $\left\{s \to \frac{t}{3},\ r \to -\frac{t^2}{6} + x\right\}$

In[323]:=
```
r = -Pi; pac =
 ParametricPlot3D[Evaluate[sol1[s, r]], {s, 0, 15}, PlotStyle → ({Thickness[0.01], Blue}),
  BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
ics = ParametricPlot3D[Evaluate[{x0, 0, Cos[x0]}], {x0, -7, 7},
  PlotStyle → ({Thickness[0.01], Green}), BoxRatios → {2, 2, 1},
  PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
ic = Graphics3D[{PointSize[0.025], Green, Point[{r, 0, Cos[r]}]}];
Show[vecs1, pac, ics, ic]
```

Out[324]=



In[325]:=
```
r = -Pi / 2; pts = Table[sol1[s, r], {s, 0, 15, .2}];
lvecs = Graphics3D[{Thickness[0.007],
    Arrow[{#, # + 1/1.6 {#[[2]], 3, -#[[3]]}}] & /@ pts},
   BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}
  ];
pac =
  ParametricPlot3D[Evaluate[sol[s, r]], {s, 0, 15}, PlotStyle → ({Thickness[0.01], Blue}),
   BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
ics = ParametricPlot3D[Evaluate[{x0, 0, Cos[x0]}], {x0, -7, 7},
  PlotStyle → ({Thickness[0.01], Green}), BoxRatios → {2, 2, 1},
  PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
ic = Graphics3D[{PointSize[0.025], Green, Point[{r, 0, Cos[r]}]}];
Show[pac, lvecs, ics, ic]
```

Out[328]=



In[329]:=
```
Clear[r];
Manipulate[ pts = Table[sol1[s, r], {s, 0, 15, .2}];
 lvecs = Graphics3D[{Thickness[0.007],
    Arrow[{#, # + 1/1.6 {#[[2]], 3, -#[[3]]}}] & /@ pts},
   BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}
  ]; pac = ParametricPlot3D[Evaluate[sol1[s, r]], {s, 0, 15},
   PlotStyle → ({Thickness[0.01], Blue}), BoxRatios → {2, 2, 1},
   PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
 ics = ParametricPlot3D[Evaluate[{x0, 0, Cos[x0]}], {x0, -7, 7},
   PlotStyle → ({Thickness[0.01], Green}), BoxRatios → {2, 2, 1},
   PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
 ic = Graphics3D[{PointSize[0.025], Green, Point[{r, 0, Cos[r]}]}];
 Show[pac, lvecs, ics, ic], {r, -2 Pi, 2 Pi, .1}]
```

Out[330]=



In[331]:=
```
Clear[r];
Manipulate[ pacm = ParametricPlot3D[Evaluate[Table[sol1[s, rr], {rr, -2 Pi, r, .2}]],
    {s, 0, 15}, PlotStyle → {{Thickness[0.005], Blue}}, BoxRatios → {2, 2, 1},
    PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
 ics = ParametricPlot3D[Evaluate[{x0, 0, Cos[x0]}], {x0, -7, 7},
    PlotStyle → ({Thickness[0.01], Green}), BoxRatios → {2, 2, 1},
    PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
 ic = Graphics3D[{PointSize[0.025], Green, Point[{r, 0, Cos[r]}]}];
 Show[pacm, ics, ic], {r, -2 Pi, 2 Pi, .1}]
```

Out[332]=



In[333]:=
```
pic = ParametricPlot3D[Evaluate[Table[sol1[s, r], {r, -15, 7, .2}]],
   {s, 0, 5}, PlotStyle → ({Thickness[0.01], #} & /@ ColorData[3, "ColorList"]),
   BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}]; Show[pic, ics]
```

Out[333]=



In[334]:=
```
uu1[x_, t_] = FullSimplify[
    (sol1[s, r][[3]]) /. Solve[{x == sol1[s, r][[1]], t == sol1[s, r][[2]]}, {s, r}]][[1]]
```

Out[334]= $e^{-t/3} \mathrm{Cos}\left[\frac{1}{6}\left(t^2 - 6\,x\right)\right]$

In[335]:=
```
Simplify[{D[uu1[x, t], x], D[uu1[x, t], t], -1}.{t, 3, -uu1[x, t]}]
```

Out[335]= 0

In[336]:=
```
gra1 = Plot3D[uu1[x, t], {x, -2 Pi, 2 Pi}, {t, 0, 10},
   PlotRange → {-1.1, 1.1}, ClippingStyle → False, PlotPoints → 100]
```

Out[336]=



In[337]:=
```
Show[vecs1, gra1]
```

Out[337]=



In[338]:=
```
Clear[r];
Manipulate[ pac =
  ParametricPlot3D[Evaluate[sol1[s, r]], {s, 0, 15}, PlotStyle → ({Thickness[0.01], Blue}),
   BoxRatios → {2, 2, 1}, PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
 ics = ParametricPlot3D[Evaluate[{x0, 0, Cos[x0]}], {x0, -7, 7},
   PlotStyle → ({Thickness[0.01], Green}), BoxRatios → {2, 2, 1},
   PlotRange → {{-2 Pi, 2 Pi}, {0, 10}, {-1.1, 1.1}}];
 ic = Graphics3D[{PointSize[0.025], Green, Point[{r, 0, Cos[r]}]}];
 Show[gra1, pac, ics, ic], {r, -2 Pi, 2 Pi, .1}]
```

Out[339]=



In[340]:=
```
Manipulate[
 Plot[uu1[x, t], {x, -2 Pi, 2 Pi}, PlotRange → {{-2 Pi, 2 Pi}, {-1.1, 1.1}}], {t, 0, 10, .1}]
```

Out[340]=



Example 3

Solve the following PDE

$u\,\frac{\partial u}{\partial x}+\frac{\partial u}{\partial t}=0,\quad u(x,0)=f(x).$

$f(x)=\arctan x,\quad f(x)=-\arctan x,\quad f(x)=e^{-x^2}.$

The vector field defined by this equation is $<z, 1, 0>$. Let us picture this field

In[341]:=
```
vecs3 = VectorPlot3D[{z, 1, 0}, {x, -2, 2}, {t, 0, 10}, {z, -1.1, 1.1},
  BoxRatios → {2, 2, 1}, PlotRange → {{-2, 2}, {0, 4}, {-1.1, 1.1}}]
```

Out[341]=



- **ArcTan[x]**

In[342]:=
```
Clear[sol3];

sol3[s_, r_] =
 FullSimplify[{x[s], t[s], z[s]} /. DSolve[{x'[s] == z[s], t'[s] == 1, z'[s] == 0, x[0] == r,
      t[0] == 0, z[0] == ArcTan[r]}, {x[s], t[s], z[s]}, s][[1]]]
```

Out[343]= {r + s ArcTan[r], s, ArcTan[r]}

In[344]:=
```
ParametricPlot[Evaluate[Table[sol3[s, r][[{1, 2}]], {r, -10, 10, .1}]],
 {s, -4, 10}, PlotRange → {{-2 Pi, 2 Pi}, {-1, 6}}, AspectRatio → Automatic]
```

Out[344]=



In[345]:=
```
ParametricPlot3D[Evaluate[sol3[s, r]], {s, 0, 4}, {r, -4, 4},
 BoxRatios → {2, 2, 1}, PlotRange → {{-4, 4}, {0, 4}, {-2.1, 2.1}}]
```

Out[345]=



In[346]:=
```
Solve[{x == sol3[s, r]〚1〛, y == sol3[s, r]〚2〛}, {s, r}]
```

Solve::nsmet : This system cannot be solved with the methods available to Solve. ≫

Out[346]=
```
Solve[{x == r + s ArcTan[r], y == s}, {s, r}]
```

In[347]:=
```
Manipulate[
 ParametricPlot[sol3[s, r]〚{1, 3}〛, {r, -2 Pi, 2 Pi}, PlotRange → {{-4, 4}, {-2, 2}}],
 {s, 0, 4, .1}]
```

Out[347]=



■ -ArcTan[x]

In[348]:=
```
Clear[sol1a];

sol3a[s_, r_] =
 FullSimplify[{x[s], t[s], z[s]} /. DSolve[{x'[s] == z[s], t'[s] == 1, z'[s] == 0, x[0] == r,
      t[0] == 0, z[0] == -ArcTan[r]}, {x[s], t[s], z[s]}, s][[1]]]
```

Out[349]= {r - s ArcTan[r], s, -ArcTan[r]}

In[350]:=
```
ParametricPlot[Evaluate[Table[sol3a[s, r][[{1, 2}]], {r, -10, 10, .1}]],
 {s, 0, 10}, PlotRange → {{-3, 3}, {0, 3}}, AspectRatio → 1 / GoldenRatio]
```

Out[350]=



In[351]:=
```
ParametricPlot3D[Evaluate[sol3a[s, r]], {s, 0, 6}, {r, -4 , 4},
 BoxRatios → {2, 2, 1}, PlotRange → {{-3, 3}, {0, 3}, {-2, 2}}]
```

Out[351]=



In[352]:=
```
Manipulate[
 ParametricPlot[sol3a[s, r][[{1, 3}]], {r, -10, 10}, PlotRange → {{-3, 3}, {-Pi / 2, Pi / 2}}],
 {s, 0, 2, .1}]
```

Out[352]=


In[353]:=
```
sol3a[s, r][[{1, 3}]]
```

Out[353]= {r - s ArcTan[r], -ArcTan[r]}

In[354]:=
```
D[sol3a[s, r][[{1, 3}]], r]
```

Out[354]= $\left\{1 - \frac{s}{1 + r^2}, -\frac{1}{1 + r^2}\right\}$

In[355]:=
```
D[sol3a[s, r][[{1, 3}]], r] /. {r → 0}
```

Out[355]= {1 - s, -1}

In[356]:=
```
ParametricPlot3D[Evaluate[sol3a[s, r]], {s, 0, 1}, {r, -4, 4}, BoxRatios → {2, 2, 1},
 PlotRange → {{-3, 3}, {0, 1}, {-Pi / 2, Pi / 2}}, PlotPoints → {100, 100}]
```

Out[356]=


- **Exp[$-x^2$]**

In[357]:=
```
Clear[sol3b];

sol3b[s_, r_] =
 FullSimplify[{x[s], t[s], z[s]} /. DSolve[{x'[s] == z[s], t'[s] == 1, z'[s] == 0, x[0] == r,
      t[0] == 0, z[0] == Exp[-r^2]}, {x[s], t[s], z[s]}, s][[1]]]
```

Out[358]= $\left\{r + e^{-r^2}\, s,\ s,\ e^{-r^2}\right\}$

In[359]:=
```
ParametricPlot[Evaluate[Table[sol3b[s, r][[{1, 2}]], {r, -4, 4, .1}]],
 {s, 0, 3}, PlotRange → {{-3, 3}, {0, 2}}, AspectRatio → 1 / GoldenRatio]
```

Out[359]=



In[360]:=
```
ParametricPlot3D[Evaluate[sol3b[s, r]], {s, 0, 3}, {r, -3 , 3},
 BoxRatios → {2, 2, 1}, PlotRange → {{-3, 3}, {0, 3}, {0, 1.2}}, PlotPoints → {100, 100}]
```

Out[360]=



In[361]:=
```
Manipulate[
 ParametricPlot[sol3b[s, r][[{1, 3}]], {r, -3, 3}, PlotRange → {{-3, 3}, {0, 1.2}}],
 {s, 0, 3, .1}]
```

Out[361]=



In[362]:=
```
sol3b[s, r][[{1, 3}]]
```

Out[362]= $\{r + e^{-r^2} s, e^{-r^2}\}$

In[363]:=
```
D[sol3b[s, r][[{1, 3}]], r]
```

Out[363]= $\{1 - 2 e^{-r^2} r s, -2 e^{-r^2} r\}$

When is $1 - 2 e^{-r^2} r s = 0$?

When is $2 e^{-r^2} r s = 1$?

Explore the family of functions $2 e^{-r^2} r s$, where s is a parameter.

In[364]:=
```
Manipulate[Plot[2 e^(-r^2) r s, {r, -3, 3}, PlotRange → {-2, 2}], {s, 0, 2}]
```

Out[364]=



In[365]:=
```
FullSimplify[D[2 e^(-r^2) r s, r]]
```

Out[365]= $e^{-r^2} (2 - 4 r^2) s$

Thus $2 e^{-r^2} r s$ has the maximum at $r = 1/\sqrt{2}$. What is the maximum?

In[366]:= $\left(2\, e^{-r^2}\, r\, s\right)$ /. $\left\{r \to \frac{1}{\sqrt{2}}\right\}$

Out[366]= $\sqrt{\frac{2}{e}}\; s$

When will $\sqrt{\frac{2}{e}}\; s = 1$?

Clearly for $s = \sqrt{\frac{e}{2}}$.

In[367]:= `ParametricPlot3D[Evaluate[sol3b[s, r]], {s, 0, 2}, {r, -3, 3}, BoxRatios → {2, 2, 1},`
`PlotRange → {{-3, 3}, {0,` $\sqrt{\frac{E}{2}}$`}, {0, 1.2}}, PlotPoints → {100, 100}]`

Out[367]=